Overall Task

Measure actual network throughput as a means to troubleshoot access database performance issues.
Comments on This Document

This is constructed in MS Word.   Lines of code may wrap at the end of a line.  Click the "Show/Hide" button [image: image1.png]


 on the Standard toolbar so that you can see where the carriage returns are located.
Background
My database uses a front-end/back-end structure to give medical reporting and tracking capability to the user.  When a patient arrives, you click a "register" button which creates records in a number of relationally connected tables.  This register function also reviews the patient's prior medical history, and turns whatever field choices it finds into a plain English, paragraph-formatted medical history report.  
The work done when the user clicks the Register button involves lots of back-and-forth trips to the back-end data files.  Consequently, it is very network intensive, and suffers serious speed degradation if the LAN is performing poorly.

This code was developed to help identify if any such LAN problems exist.

Initial Code Requirements

You need to know where the local front-end MDE files live.  This is one directory to which you can be sure the user will have read/write/create/delete permissions.  If the user lacked these permissions, the MDE file would be unable to create an LDB file successfully.
You also need to know where the back-end MDB file(s) live.  This is the directory you are connecting to across the network.

Where is the front-end directory?

My application file is called MBsProg.mde.  The filename itself has 11 characters in it.  So to find the local front-end directory name, use this:
Dim vCurrentDB as string

Dim vCurrentDBpath as string

vCurrentDB = CurrentDb.Name

vCurrentDBpath = Left(vCurrentDB, Len(vCurrentDB) - 11)

Where is the back-end directory?
t100-Clients is one of my table names, that exists in a back-end file called MBsData.mdb.  I can find out where it lives with this code.  Note that as with the front end, the back-end filename has 11 characters.  So to get the folder name, I need to subtract 11 characters from the path that my function returns.
Dim vBackEndPath as string

vBackEndPath = GetAttachedPath("","t100-Clients")
vBackEndPath = Left(vBackEndPath, Len(vBackEndPath)-11)
'''''''''''''''''''''''''''''''''''''''''''''''''
Function GetAttachedPath(strDatabase As String, strTable As String) As String

' Comments  : returns the attached path of the named table

' Parameters: strDatabase - name and path of database to look in or "" (blank string) for the current database

'             strTable - name of table to get the path for

' Returns   : path

'

          Dim dbsTmp As Database

          Dim tdfTmp As TableDef

          Dim strConnect As String

          Dim intPos As Integer

  On Error GoTo ERR_GetAttachedPath

14030        If strDatabase = "" Then

14040           Set dbsTmp = CurrentDb()

14050        Else

14060          Set dbsTmp = DBEngine.Workspaces(0).OpenDatabase(strDatabase)

14070        End If

14080        Set tdfTmp = dbsTmp.TableDefs(strTable)

14090        strConnect = tdfTmp.Connect

14100        If strConnect <> "" Then

14110          intPos = InStr(strConnect, "=")

14120          GetAttachedPath = Mid$(strConnect, intPos + 1, Len(strConnect))

14130          End If

14140         dbsTmp.Close

Exit_GetAttachedPath:

14150           Set dbsTmp = Nothing

14160           Set tdfTmp = Nothing

14170         Exit Function

ERR_GetAttachedPath:

14180         GetAttachedPath = ""

14190

        Dim ErrorMsg1 As String, vResult As Boolean

    ErrorMsg1 = "MammoBase has encountered a problem on line " & Erl & " of:" & vbCrLf & vbCrLf & vbTab & _

    "FunctionsAtoM" & vbCrLf & vbCrLf & "(Problem type #" & Err.Number & ": " & Err.Description & ")" 

14200       msgbox ErrorMsg1

14230         Resume Exit_GetAttachedPath

End Function
Create A File Of A Known Size, And Copy It From The Front End-Folder To The Back-End Folder
I have a button they can click at any time that will execute a function to assess their network performance.  In addition, I have a hidden form that opens when my front-end opens.  I have a timer function on that form that runs this code once an hour.
The objective here is to determine not only the network performance at this instant, but to get a picture of how it changes over time.  If network speed is regularly 35 mbps and then drops at a certain time of day down to 3 mbps, it will be thought-provoking for the local LAN specialist.
Public Function NetworkTestLog()

On Error GoTo ErrorHandler

  Dim vPath As String, i As Long

  Dim vFileName As String

  Dim vStartTime As Single

  Dim vEndTime As Single

  Dim vTotalTime As Single

  Dim strDataPath As String

  Dim vLengthFilePath As Long

  Dim vTargetPath As String

  Dim vFileSize As Long

  Dim vPerformance As Single

  Dim vPercent As Single

  Dim vSource As String

  Dim vTarget As String

  Dim rst As Recordset

  Dim FS

  Dim vTime As String

  Dim rst800 As Recordset

150     Set FS = CreateObject("Scripting.FileSystemObject")

160     Set rst800 = CurrentDb.OpenRecordset("t800-Variables")

170

180    vReturn = SysCmd(acSysCmdSetStatus, "Running network diagnostic...")

240 
290   vSource = vCurrentDBpath
300   vTarget = vBackEndPath
310   vFileName = CStr(Timer) & ".test"

320   vTargetPath = strDataPath & vFileName

330   vFileName = vSource & vFileName

350   Open vFileName For Output As #2

360     For i = 1 To 16384  'i.e. a 512 kb test file is created

370       Print #2, "123456789012345678901234567890"

380     Next i

390   Close #2

400   'The for/next loop above takes 0.171875 seconds to complete on

        '2 mb file.  for i = 1 to 16384 will give a 0.5 MB file.

410   'I don't think I need to insert a doEvents line.

420   'Inserting a DoEvents line just after "Print #2..." stretches this process out

430   'to 240.5781 seconds (i.e. 4 minutes).  It does mean that I can continue to

440   'use MammoBase and notice no performance hit.

450   vStartTime = Timer

460   FileCopy vFileName, vTargetPath

470   vEndTime = Timer

480   vTotalTime = vEndTime - vStartTime

490   If vTotalTime = 0 Then vTotalTime = 0.01

510   vReturn = SysCmd(acSysCmdSetStatus, "Test Complete")
520   vFileSize = 4194304 ' bytes * 8 = bits

530   vPerformance = vFileSize / vTotalTime

540   vPerformance = vPerformance / 1048576 'bps / 1048576 = mbps

550   vPercent = Format(vPerformance, "0.#")

560   vSource = vSource & "*.test"

570   vTarget = vTarget & "*.test"

580   Kill vSource

590   Kill vTarget

600   Set rst = CurrentDb.OpenRecordset("t925-NetworkPerformance")

610   If rst.RecordCount > 0 Then

620    rst.MoveLast

630    rst.MoveFirst

640    Do Until rst.RecordCount < 24


'I keep a maximum of 24 records in this table... 



'about 3 days worth of hourly tests.

650        rst.Delete

660        rst.MoveLast

670        rst.MoveFirst

680    Loop

690   End If
'All of my vPerformance estimates are based on testing done on my own network, using 3 PCs: my source, my target, and a 3rd PC.  I would load up the network by getting my target PC to initiate the copy of a massive file from itself to the 3rd PC, then set it to work zipping another big file.

'This measurement tool measures not just raw network speed, but it is affected as well by CPU load on both the target and source PCs.

'By loading up the CPU, I could simulate a slower network.

'Once I did this simulation, I did some actual performance testing with my application on a common task:  the time required to register a patient for a mammogram.
'Based on my results, I was able to project how long such a

'task would take the end user.  I am trying to translate

'mbps (which most medical end users don't understand) into

'something they DO understand:  how long it takes to do

'this or that function in the application.

'

'Hence, the vPerformance lines of code immediately below.
700   If vPerformance < 1 Then vTime = "30 sec+"

710  If vPerformance >= 1 And vPerformance < 1.5 Then vTime = " 12 sec" ' 1 mbps

720  If vPerformance >= 1.5 And vPerformance < 2.5 Then vTime = " 11 sec" ' 2 mbps

730  If vPerformance >= 2.5 And vPerformance < 3.5 Then vTime = " 10 sec" ' 3 mbps

740  If vPerformance >= 3.5 And vPerformance < 4.5 Then vTime = "  8 sec" ' 4 mbps

750  If vPerformance >= 4.5 And vPerformance < 10 Then vTime = "  5 sec" ' 5-10 mbps

760  If vPerformance >= 10 And vPerformance < 20 Then vTime = "2.70 sec" ' 10-20 mbps

770  If vPerformance >= 20 And vPerformance < 40 Then vTime = "0.90 sec" ' 20-40 mbps

780  If vPerformance >= 40 And vPerformance < 60 Then vTime = "0.70 sec" ' 40-60 mbps

790  If vPerformance >= 60 Then vTime = "0.50 sec" ' 60+ mbps

800

810    If rst.RecordCount > 0 Then rst.MoveLast

820    rst.AddNew

830      rst!Date = Date

840      rst!time = Format(Now, "Short Time")

850      rst!mbps = vPerformance

860      rst!performance = vPerformance * 0.01

870      rst!TimeToRegister = vTime

920    rst.Update

930

ExitRoutine:

    If Not rst Is Nothing Then rst.Close

    Set rst = Nothing

    If Not rst800 Is Nothing Then rst800.Close

    Set rst800 = Nothing

    Set FS = Nothing

    Close #1

    Close #2

    Exit Function

ErrorHandler:

    If Err.Number = 76 Or Err.Number = 53 Then

        Resume ExitRoutine

    End If

    Dim ErrorMsg1 As String, vResult As Boolean

    ErrorMsg1 = "MammoBase has encountered a problem on line " & Erl & " of:" & vbCrLf & vbCrLf & vbTab & "FunctionsNtoZ.NetworkTestLog" & vbCrLf & vbCrLf & "(Problem type #" & Err.Number & ": " & Err.Description & ")"  

     msgbox ErrorMsg1
    Resume ExitRoutine

End Function
You will notice in line 600, I write my data to a table.  Here is the structure of the table.

Output Table Info to a TXT File That May Be Printed
I have this routine attached to a button that gives me version information, as well as info on available hard drive space, customer address, etc.

I will try to strip out everything from this routine that is not relevant to performance testing.

Private Sub cmdMammoBaseVersion_Click()

On Error GoTo Err_cmdMammoBaseVersion_Click

  Dim vPath As String

  Dim FS, f, vFileName, cr As String

  Dim vNum As Long, vStr As String, vShell As String

  Dim vTmpStr As String

  Dim vTmpLength As Long

  Dim rst As Recordset

  Dim vMBPS As String

  Dim vPerformance As String

10998   NetworkTestLog

'I am going to save this network test out to a log file.  I go to some effort to ensure I have a unique filename.

11600   Set FS = CreateObject("Scripting.FileSystemObject")

11610   vNum = 1

11620   cr = vbCrLf & vbCrLf

11630   vPath = vErrorLogPath

11640    If Right(vPath, 1) <> "\" Then vPath = vPath & "\"

11650   vFileName = vPath & "Status_" & Format(Date, "mmddyy") & "_" & vNum & ".log"

11660   If FS.FolderExists(vPath) = False Then FS.createfolder (vPath)

11670

11680 TryAgain:

11690   If FS.FileExists(vFileName) = True Then

11700       vNum = vNum + 1

11710       vFileName = vPath & "Status_" & Format(Date, "mmddyy") & "_" & vNum & ".log"

11720       If vNum >= 1000 Then

11730           MsgBox "There appear to be a large number of MammoBase log files in " & vPath & ".  I cannot create any more until you move or delete the existing files."

11740           Exit Sub

11750       End If

11760       GoTo TryAgain

11770   End If

11780

        Set rst = CurrentDb.OpenRecordset("t925-NetworkPerformance")

        vStr = vStr & " Date     Time     Measured Network     Actual vs       Time to Register" & vbCrLf

        vStr = vStr & "                      Throughput      Rated Network       One Patient" & vbCrLf

        vStr = vStr & "                        (mbps)         Performance        (estimated)" & vbCrLf

        Do Until rst.EOF

            If rst!mbps < 10 Then

                vMBPS = " " & Format(rst!mbps, "0.00")

            ElseIf rst!mbps > 99 Then

                vMBPS = ">100"

            Else

                vMBPS = Format(rst!mbps, "0.00")

            End If

            If (rst!performance * 100) < 10 Then

                vPerformance = " " & Format((rst!performance * 100), "0") & "%"

            ElseIf (rst!performance * 100) > 99 Then

                vPerformance = ">100%"

            Else

                vPerformance = Format((rst!performance * 100), "0") & "%"

            End If

            vStr = vStr & Format(rst!Date, "mm/dd/yy") & "  " & _

                Format(rst!time, "Short Time") & "         " & _

                vMBPS & "              " & _

                vPerformance & "              " & rst!TimeToRegister & vbCrLf

            rst.MoveNext

        Loop

        vStr = vStr & vbCrLf & "If measured mbps is lower than rated bandwidth, it may be due to dropped packets due to half/full/auto duplex mismatch between switch and any of the NICs, CPU load levels on both the local box and the server, or a general inability of select components to saturate the data pipe.  Locally stored data (or gigabit Ethernet) may display performance greater than 100%."

12050   Set f = FS.CreateTextFile(vFileName, True)

12060   f.writeline (vStr)

12070   f.Close

12080   vShell = "Notepad.exe " & vFileName

12090   vReturn = Shell(vShell, vbMaximizedFocus)

Exit_cmdMammoBaseVersion_Click:

    Set f = Nothing

    Set FS = Nothing

    If Not rst Is Nothing Then rst.Close

    Set rst = Nothing

    Exit Sub

Err_cmdMammoBaseVersion_Click:

    Dim ErrorMsg1 As String, vResult As Boolean

    ErrorMsg1 = "MammoBase has encountered a problem on line " & Erl & " of:" & vbCrLf & vbCrLf & vbTab & "Form_f0-Master.cmdMammoBaseVersion_Click" & vbCrLf & vbCrLf & "(Problem type #" & Err.Number & ": " & Err.Description & ")"  

    vResult = msgbox(ErrorMsg1)

    Resume Exit_cmdMammoBaseVersion_Click

End Sub
